with Text_IO; use Text_IO;

procedure EndlicheAutomaten is

   type detueber is

     array (Integer range <>,

            Character range <>) of Integer;

   -- ordnet einem Zustand und einem Zeichen

   -- einen Nachfolgezustand zu

   type det_ueber_type (n : Positive) is record

      ueber : detueber (0..n,' ' .. '~');

   end record;

   -- Einschraenkung auf Zustaende 0..n-1 und Fangzustand n

   -- (damit die Uebergangsfunktion total ist);

   -- T ist die Menge der druckbaren ASCII-Zeichen

   type charFktn is

     array (Integer range <>) of Boolean;

   -- Charakteristische Funktion fuer Mengen

   -- ganzer Zahlen

   type charFktn_type (n : Positive) is record

      inh : charFktn (0..n) := (others => False);

   end record;

   -- beschraenkt die Obermenge auf die Zustaende

   -- 0 bis n;

   -- Initialisierung beschreibt die leere Menge

   type DEA_type (n : Positive) is record

      det : det_ueber_type (n);

      F   : charFktn_type (n);

   end record;

   -- beschreibt einen deterministischen endlichen

   -- Automaten mit

   --   Q = {0,1,...,n-1,n} (n ist Fangzustand)

   --   q0 = 0

   --   T = Menge der druckbaren ASCII-Zeichen

   --   delta = det

   --   F = F

   DEA1 : DEA_type (4);

   procedure pruefe (DEA : DEA_type; w : String) is

      akt : Integer := 0; -- aktueller Zustand

   begin

      for i in 1 .. w'last loop

         akt := DEA.det.ueber (akt, w(i));

      end loop;

      put ("""" &w& """" & " gehoert");

      if not DEA.F.inh(akt) then

         put (" nicht");

      end if;

      put (" zum Sprachschatz");

      new_line(2);

   end pruefe; 

   type DEA_ref is access DEA_Type;

   -- nuetzlich, wenn man z.B. mehrere DEAs in einem

   -- Array speichern moechte (siehe Scanner!)
begin

   DEA1.det.ueber :=

     (0 => ('a' => 1, 'b' => 2, others => 4),

      1 => ('a' => 0, 'b' => 3, others => 4),

      2 => ('a' => 3, 'b' => 0, others => 4),

      3 => ('a' => 2, 'b' => 1, others => 4),

      4 => (others => 4));

   DEA1.F.inh := (3 => True, others => False);

   pruefe (DEA1, "a");

   pruefe (DEA1, "abb");

   pruefe (DEA1, "abaa");

   pruefe (DEA1, "abbb");

   pruefe (DEA1, "ab");

   pruefe (DEA1, "ab ");
   -- Ergebnis des Programmlaufs:
   -- "a" gehoert nicht zum Sprachschatz
   -- "abb" gehoert nicht zum Sprachschatz
   -- "abaa" gehoert zum Sprachschatz
   -- "abbb" gehoert zum Sprachschatz
   -- "ab" gehoert zum Sprachschatz
   -- "ab " gehoert nicht zum Sprachschatz
end EndlicheAutomaten;

